Kapitel  2.18 : Lösung der Übungsaufgaben
55. Von den vier hier vorgestellten Funktionen sind nur zwei wirklich unverzichtbar, die anderen beiden könnte man sich auch selbst schreiben. Welches sind die beiden elementaren Funktionen?

Wirklich unverzichtbar sind die beiden Funktionen (substr) und (strcat). Die Funktion (strlen) könnte man, mit dem bis hierher erworbenen Wissen, zum Beispiel so ersetzen:

(defun Ersatz-strlen (str / zähler)

  (setq zähler 0)

  (while (/= "" str)

    (setq str (substr str 2)

          zähler (1+ zähler)

    )

  )

)

Die Ersatz-Funktion für (strcase) könnte in ähnlicher Weise eine Zeichenkette durchlaufen, für jedes Buchstaben-Element den zugehörigen Groß-, bzw. Kleinbuchstaben finden (sie liegen in der ASCII-Tabelle immer 32 Zeichen weit auseinander), und am Ende die Zeichenkette mit (strcat) wieder zusammenfügen.

Für das Verschieben um 32 Zeichen verwenden wir diese Funktion, die auf die Funktionen (ascii) und (chr) vorgreift:

(defun GROSS->klein->GROSS (zeichen modus)

  (chr (modus (ascii zeichen) 32))

)

Zeichen ist das Zeichenkettenelement und modus gibt die Verschiebung in positiver (+) oder negativer (-) Richtung an.

Und hier die Funktion Ersatz-strcase, die mit dem bisher Bekannten erzeugt werden kann:

(defun Ersatz-strcase (str klein? / ergebnis zeichen)

  (setq ergebnis "")

  (while (/= "" str)

    (setq zeichen (substr str 1 1)

          str     (substr str 2)

    )

    (setq zeichen

           (cond

             ((and

                klein?

                (<= 65 (ascii zeichen) 90)

              )

              (GROSS->klein->GROSS zeichen +)

             )

             ((<= 97 (ascii zeichen) 122)

              (GROSS->klein->GROSS zeichen -)

             )

             ('T zeichen)

           )

    )

    (setq ergebnis (strcat ergebnis zeichen))

  )

  ergebnis

)

Der einzige Unterschied zu (strcase) ist die zwingend notwendige Übergabe des Arguments für die Kleinschreibung.

56. Definieren Sie eine Funktion (str-first), die das erste Zeichen eines Strings zurückgibt.

(defun str-first (str)

  (substr str 1 1)

)

57. Definieren Sie eine Funktion (str-last), die das letzte Zeichen eines Strings zurückgibt.

(defun str-last (str)

  (substr str (strlen str))

)

58. Definieren Sie eine Funktion (str-pos), die feststellt, ob in einer Zeichenkette (1. Argument) ein bestimmtes Zeichen vorkommt (2. Argument). Die Funktion soll, wenn das Zeichen nicht vorkommt, 0 zurückgeben. Wenn es gefunden wurde, wird die Position retourniert.

(defun str-pos (str zeichen / i)

  (setq i 1)

  (while

    (and

      (<= i (strlen str))

      (/= (substr str i 1) zeichen)

      )

     (setq i (1+ i))

     )

  (if (<= i (strlen str))

    i

    0

    )

  )

59. Definieren Sie eine Funktion, die mit der Syntax (str-insert <zeichenkette> <neue-zeichenkette> <position>) ein oder auch mehrere Zeichen in eine Zeichenkette an der Stelle position einfügt.

(defun str-insert (zeichenkette neue-Zeichenkette position /)

  (strcat

    (substr zeichenkette 1 (1- position))

    neue-Zeichenkette

    (substr zeichenkette position)

  )

)

60. Definieren Sie eine Funktion (str-reverse <str>), die eine Zeichenkette rückwärts zurückgibt.

(defun str-reverse (str / ergebnis)

  (setq ergebnis (substr str 1 1))

  (while (/= "" str)

    (setq str      (substr str 2)

          ergebnis (strcat

                     (substr str 1 1)

                     ergebnis

                   )

    )

  )

)

